Universidade de São Paulo

Instituto de Física de São Carlos

Introdução à Física Computacional

—

Projeto 4

Levy Bruno do Nascimento Batista − 11212550

Prof. Francisco Castilho Alcaraz

São Carlos

11/2021 − 12/2021

1 Tarefa A

Nessa primeira tarefa, foi estudado o movimento de um pêndulo simples a partir de dois métodos diferentes de discretização do tempo, analisando os comportamentos de $\theta(t)$ e $E(t)$ de cada um, e assim sendo possível estabelecer qual é o mais adequado. Os métodos em questão são o de Euler e o de Euler-Cromer, cuja diferença está na iteração dos valores de ângulo.

No código abaixo, cada método foi implementado em uma subrotina diferente, e as condições iniciais para ambas situações são as mesmas, isto é, $\theta_0 = 15°$ e $\omega_0 = 0$, e a seguir estão os gráficos das simulações para 10 períodos. Em cada subrotina, um arquivo com os instantes de tempo e os respectivos ângulo e energia do sistema é gerado.

tarefa-A-11212550.f:

```fortran
      program tarefaA

      implicit real*8 (a-h,o-z)
c     pi, comp e g serão utilizadas nas subrotinas implementadas
      common /cte/pi, comp, g
      pi = dacos(-1.d0)
      comp = 9.8d0
      g = 9.8d0

c     subrotina que calcula theta e E pelo método de Euler
      theta = pi/12
      omega = 0.d0

      call euler(theta, omega)

c     subrotina que calcula theta e E pelo método de Euler-Cromer
      theta = pi/12
      omega = 0.d0

      call euler_cromer(theta, omega)

      end

      subroutine euler (ang, vel)

        implicit real*8 (a-h,o-z)
        common /cte/pi, comp, g
c       intervalo utilizado na discretização do tempo
        deltat = 0.01d0
c       tempo total de simulação
        T = 0.d0

```

```
        open(1, file='1saida-A-11212550')
        write(1, 100) 't(s)', '|', 'theta(rad)', '|', 'E(J)'
        E = (comp*vel)**2/2 + g*comp*(1.d0 - dcos(ang))
        write(1, 200) T, '|', ang, '|', E
 10     if (T.le.20*pi) then
          velaux = vel
c         itera os valores de theta e omega
          vel = vel - ang*deltat
          ang = ang + velaux*deltat
c         calcula o novo valor de energia do sistema
          E = (comp*vel)**2/2 + g*comp*(1.d0 - dcos(ang))
c         registra a passagem do tempo
          T = T + deltat
          write(1, 200) T, '|', ang, '|', E
          goto 10
        end if
        close(1)

 100    format(A5, A, A10, A, A4)
 200    format(F5.2, A, F10.5, A, F9.6)

      end

      subroutine euler_cromer (ang, vel)

        implicit real*8 (a-h,o-z)
        common /cte/pi, comp, g
c       intervalo utilizado na discretização do tempo
        deltat = 0.01d0
c       tempo total de simulação
        T = 0.d0

        open(1, file='2saida-A-11212550')
        write(1, 100) 't(s)', '|', 'theta(rad)', '|', 'E(J)'
        E = (comp*vel)**2/2 + g*comp*(1.d0 - dcos(ang))
        write(1, 200) T, '|', ang, '|', E
 10     if (T.le.20*pi) then
c         itera os valores de theta e omega
          vel = vel - ang*deltat
          ang = ang + vel*deltat
c         calcula o novo valor de energia do sistema
          E = (comp*vel)**2/2 + g*comp*(1.d0 - dcos(ang))
c         registra a passagem do tempo
          T = T + deltat
          write(1, 200) T, '|', ang, '|', E
          goto 10
        end if
        close(1)

 100    format(A5, A, A10, A, A4)
```

```
200     format(F5.2, A, F10.5, A, F9.6)

      end
```

O programa gerou os seguintes resultados para o método de Euler:

Figura 1: Ângulo em relação a vertical por tempo.



De início, observa-se a inconsistência entre o resultado exposto na Figura 1 e o esperado pela teoria, uma vez que a amplitude do pêndulo está crescendo com o tempo. Essa inconsistência fica mais evidente no gráfico de $E(t)$.

Figura 2: Energia do sistema por tempo.



O gráfico da Figura 2 evidencia que o método de Euler não é adequado para estudar esse problema, pois ele leva a energia crescer indefinidamente, não respeitando o princípio da coservação de energia.

Agora, segue os resultados obtidos pelo método de Euler-Cromer:

Figura 3: Ângulo em relação a vertical por tempo.



Dessa vez, a amplitude de oscilação permanece a mesma, o que era esperado pela teoria.

Figura 4: Energia do sistema por tempo.



Já a energia se comporta de forma diferente da esperada, uma vez que ela está oscilando no tempo, mas a amplitude é baixa em relação ao valor máximo que ela alcança e a mesma não diverge, como acontece com o método anterior, logo o método de Euler-Cromer gera resultados mais próximos do teórico.

2 Tarefa B

Na segunda tarefa, nas duas primeiras partes foram estudadas a dependência do período com o ângulo inicial de acordo com o método de Euler-Cromer sem aproximar para ângulos pequenos, com o resultado numérico da integral elíptica presente no roteiro e da fórmula com a aproximação em segunda ordem em θ também presente no roteiro; na terceira parte, é acrescentado o termo de amortecimento no estudo de oscilações e na quarta o termo associado à força externa. Nesse último são traçados gráficos de ângulo e velocidade angular em função do tempo para 3 valores diferentes de amplitude de força externa, sendo possível observar qual a resposta do sistema para diferentes estímulos.

No código abaixo, a subrotina B1 está relacionada com o itens B1 e B2, enquanto a B3 está com o item B3 e a B4 com o item B4 do roteiro. Em todas elas são gerados arquivos de saída com dados importantes para a compreensão do problema.

tarefa-B-11212550.f:

```
      program tarefaB

      implicit real*8 (a-h,o-z)
c     pi foi definido dessa forma pois será usado nas subrotinas
      common /cte/pi
      pi = dacos(-1.d0)

c     as três subrotinas são chamadas em sequência
      call B1()
      theta = pi/18
      omega = 0.d0
      call B3(theta, omega)
      theta = pi/18
      omega = 0.d0
      call B4(theta, omega)

      end

      subroutine B1 ()

        implicit real*8 (a-h,o-z)
        common /cte/pi
c       intervalo utilizado na discretização do tempo
        deltat = 0.01d0
```

```
c       ângulo inicial, que vai sendo alterado até pi/3
        ang = pi/36
c       epsilon utilizado no cálculo da integral elíptica
        e = 0.001d0
c       número de divisões do intervalo de integração por Boole
        n = 384
        open(1,file='1saida-B-11212550')
        write(1, 100) 'theta0', '|', 'T/Euler-Cromer', '|',
     +              'T/integral', '|', 'T/fórmula'

        do i = 1, 12
c         método de Euler-Cromer para encontrar o período
          angaux = ang
          vel = 0.d0
          k = 0
          T = 0.d0
 10       if (T.le.3*pi) then
            angaux2 = angaux
            vel = vel - dsin(angaux)*deltat
            angaux = angaux + vel*deltat
            T = T + deltat
c           marca a primeira vez que o ângulo mda de sinal
            if (angaux2*angaux.lt.0.and.k.eq.1) then
              Tf = T
              k = 2
            end if
c           marca a segunda vez que o ângulo muda de sinal
            if (angaux2*angaux.lt.0.and.k.eq.0) then
              Ti = T
              k = 1
            end if
            goto 10
          end if
c         cálculo da integral elíptica
          valor = 4*dsqrt(2*e/dsin(ang))
          thint = 0
          h = (ang - e)/n
c         cálculo da parte numérica pelo método de Boole
          do j = 0, (n/4 - 1)
            thint=thint +
                (2*h/45)*(7/dsqrt(dcos(4*j*h)-dcos(ang))+
     +            32/dsqrt(dcos((4*j+1)*h)-dcos(ang))+
         12/dsqrt(dcos
     +            ((4*j+2)*h)-dcos(ang))+
         32/dsqrt(dcos((4*j+3)*h)-
     +            dcos(ang))+
         7/dsqrt(dcos((4*j+4)*h)-dcos(ang)))
          end do
          thint = 2*dsqrt(2.d0)*thint
c         soma das partes analítica e numérica
```

```
          valor = valor + thint
          write(1, 200) ang*180/pi, '|', 2*(Tf - Ti), '|',
              valor,
     +                    '|', 2*pi*(1 + ang**2/16)
c         incremento do ângulo inicial
          ang = ang + pi/36
        end do
        close(1)

 100    format(A8, A, A14, A, A10, A, A10)
 200    format(F8.1, A, F14.2, A, F10.2, A, F9.2)
      end

      subroutine B3 (ang, vel)

        implicit real*8 (a-h,o-z)
        common /cte/pi
c       intervalo utilizado na discretização do tempo
        deltat = 0.01d0
c       tempo total de movimento
        T = 0.d0
c       coeficiente de amortecimento
        q = 0.5d0

        open(1, file='2saida-B-11212550')
        write(1, 100) 't(s)', '|', 'theta(rad)'
 10     if (T.le.10*pi) then
c         aplicação do método de Euler-Cromer
          write(1, 200)T, '|', ang
          vel = vel - dsin(ang)*deltat - q*vel*deltat
          ang = ang + vel*deltat
          T = T + deltat
          goto 10
        end if
        write(1, 200)T, '|', ang
        close(1)

 100    format(A9, A, A10)
 200    format(F9.2, A, F8.5)

      end

      subroutine B4 (ang, vel)

        implicit real*8 (a-h,o-z)
        common /cte/pi
c       intervalo utilizado na discretização do tempo
        deltat = 0.03d0
c       tempo total de movimento
        T = 0.d0
```

```fortran
c       coeficiente de amortecimento
        q = 0.5d0
c       amplitude da força externa
        F0 = 1.2d0
c       frequência da força externa
        omegaF = 2.d0/3

        open(1, file='3saida-B-11212550')
        open(2, file='4saida-B-11212550')
        write(1, 100) 't(s)', '|', 'theta(rad)'
        write(2, 100) 't(s)', '|', 'vel(rad/s)'
 10     if (T.le.20*pi) then
c         aplicação do método de Euler-Cromer
          write(1, 200)T, '|', ang
          write(2, 200)T, '|', vel
          vel = vel - dsin(ang)*deltat - q*vel*deltat
     +          + F0*dsin(omegaF*T)*deltat
          ang = ang + vel*deltat
          T = T + deltat
          goto 10
        end if
        write(1, 200)T, '|', ang
        write(2, 200)T, '|', vel
        close(1)
        close(2)

 100    format(A9, A, A10)
 200    format(F9.2, A, F8.5)

      end
```

2.1 B1

Nesse primeiro item, foram feitas medidas de período para 12 ângulos iniciais diferentes de duas formas distintas; a primeira pelo método de Euler-Cromer, pegando dois instantes de tempo seguidos nos quais o ângulo inverte de sinal, sendo o intervalo entre esses dois instantes igual a meio período, e a segunda resolvendo numericamente a integral elíptica:

$$T = \sqrt{\frac{2l}{g}} \int_{-\theta_0}^{\theta_0} \frac{d\theta}{\sqrt{cos\theta - cos\theta_0}}$$

de acordo com o método apresentado pelo professor. No cálculo da parte sem a singularidade foi utilizado o método de Boole, com o intervalo de integração sendo dividido em $N = 384$ partes, que mostrou ser o intervalo que gera o menor desvio para tal método, segundo Projeto 3.

`1saida-B-11212550`:

```
theta0|T/Euler-Cromer|T/integral|T/fórmula
   5.0|          6.28|      6.29|     6.29
  10.0|          6.30|      6.29|     6.30
  15.0|          6.30|      6.31|     6.31
  20.0|          6.34|      6.33|     6.33
  25.0|          6.36|      6.36|     6.36
  30.0|          6.38|      6.39|     6.39
  35.0|          6.42|      6.43|     6.43
  40.0|          6.48|      6.48|     6.47
  45.0|          6.54|      6.54|     6.53
  50.0|          6.60|      6.60|     6.58
  55.0|          6.66|      6.67|     6.65
  60.0|          6.74|      6.75|     6.71
```

Os valores obtidos por essas duas formas correspondem às duas primeiras colunas da tabela acima. Observe que há grande concordância entre os dois métodos, mesmo para ângulos iniciais grandes, o que mostra que o algoritmo de Euler-Cromer adaptado (com senθ no lugar de θ) pode ser utilizado para calcular o período de oscilação, uma vez que gera resultados muito próximos do esperado pela teoria.

2.2 B2

Ainda sobre a tabela exposta no item anterior, na última coluna estão os valores de período obtidos segundo a expressão:

$$T = 2\pi\sqrt{\frac{l}{g}(1+\frac{\theta^2}{16})}$$

Note que esses resultados estão de acordo com os outros dois métodos até $\theta_0 = 35,0°$, e a partir daí os desvios tendem a crescer conforme o ângulo inicial vai aumentando. Isso se dá porque a expressão "correta"é uma série infinita em θ, e a fórmula acima é apenas uma aproximação em segunda ordem, de forma que, para ângulos maiores, os termos de ordem mais alta começam a aparecer no resultado. Veja ainda que, para ângulos pequenos ($\theta_0 < 15°$), o resultado varia pouco, o que mostra que o período praticamente independe do ângulo inicial nesse intervalo, que corresponde ao limite das oscilações harmônicas.

2.3 B3

Agora, numa nova adaptação ao algoritmo de Euler-Cromer, o termo associado à força resistiva é adicionado. Nesse caso, e em todos no qual essa força for considerada, o coeficiente de amortecimento γ é tomado como $\gamma = 0,5$. Na subrotina B3 são gerados os dados de $\theta(t)$ para esse caso, que podem ser visualizados no gráfico da Figura 5.

Figura 5: Ângulo em relação a vertical por tempo.



Observe que o sistema se encontra no regime de amortecimento subcrítico, no qual ele oscila algumas vezes antes da força resisitiva dissipar toda energia mecânica. Isso se confirma pelo fato de que, substituindo os valores utilizados, $\frac{\gamma^2}{4} - \omega_0^2 = \frac{1}{16} - 1 < 0$, o que é esperado no regime subcrítico.

2.4 B4

Agora, além de se considerar o termo de amortecimento, temos também o termo associado a um forçamento externo. A subrotina B4 opera de forma semelhante à B3, a diferença se encontra na modificação do algoritmo de Euler-Cromer para encaixar o termo de F_0, cujo valor pode ser alterado no início do código; nesse sentido, foram construídos gráficos de $\theta(t)$ e $\omega(t)$ para 3 valores diferentes de F_0, de forma que foi possível observar diferenças significativas na resposta do sistema nos 3 casos.

Figura 6: Ângulo e velocidade angular por tempo para $F_0 = 0$.



O primeiro caso, que corresponde à $F_0 = 0$, é o mesmo estudado no item B3, com a diferença que agora podemos visualizar também o comportamento de $\omega(t)$. Ambas as curvas presentes na Figura 6 estão de acordo com o resultado esperado pela teoria para um amortecimento subcrítico.

Figura 7: Ângulo e velocidade angular por tempo para $F_0 = 0,5$.



Já para o caso que $F_0 = 0,5$, observou-se que, após um curto período inicial no qual o termo transiente prevalece, o sistema passa a oscilar com um período bem definido, dado aproximadamente por $T = 26,85 - 17,43 = 9,42$ s (esses dois instantes de tempo correspondem aos primeiros momentos no regime estacionário nos quais o ângulo é mínimo). Note que esse valor está bastante coerente com o esperado teórico, o qual é dado por $T = \frac{2\pi}{\frac{2}{3}} = 3\pi \approx 9,42$ s.

Figura 8: Ângulo e velocidade angular por tempo para $F_0 = 1,2$.



Por fim, para o caso $F_0 = 1,2$, perceba que o movimento não é mais periódico, mesmo apresentando algumas poucas oscilações no meio da simulação e a velocidade não aparente divergir no intervalo estudado, o ângulo rapidamente cresce para valores muito maiores que θ_0, em módulo (os valores de θ maiores que 2π em módulo mostram que o pêndulo chega a dar várias voltas em torno do ponto de suporte). Nesse caso, diferentemente dos dois anteriores que representam movimentos determinísticos, o sistema se comporta de maneira caótica, o que pode ser confirmado na próxima tarefa.

3 Tarefa C

Nessa terceira tarefa, foi estudado mais a fundo o movimento caótico resultante do $F_0 = 1,2$, comparando ainda com a situação determinística associada a $F_0 = 0,5$; para isso, analisou-se o comportamento do $\Delta\theta(t)$ para dois osciladores sob mesmo forçamento externo e velocidade inicial nula, com $\Delta\theta(0) = 0,001$ rad. É interessante observador como esse parâmetro evoluiu de forma diferente para cada caso, de forma que, para cada um deles, estimou-se o expoente de *Liapunov* λ a partir do gráfico de $\Delta\theta(t)$ em escala semi-logarítmica.

No código abaixo, o valor de F_0 pode ser facilmente alterado no início, junto com as condições iniciais de cada oscilador. O programa gera ainda um arquivo de saída com os dados da evolução da diferença de ângulo ao longo do tempo.

`tarefa-C-11212550.f`:

```fortran
      program tarefaC

        implicit real*8 (a-h,o-z)
        pi = dacos(-1.d0)
c       ângulos e velocidades iniciais para os dois osciladores
        ang1 = pi/18
        ang2 = ang1 + 0.001d0
        vel1 = 0.d0
        vel2 = 0.d0
c       intervalo utilizado na discretização do tempo
        deltat = 0.03d0
c       tempo total de movimento
        T = 0.d0
c       termo de amortecimento
        q = 0.5d0
c       amplitude da força externa
        F0 = 1.2d0
c       frequência da força externa
        omegaF = 2.d0/3

        open(1, file='saida-C-11212550')
        write(1, 100) 't(s)', '|', 'theta(rad)'
 10     if (T.le.200*pi) then
          write(1, 200)T, '|', ang2-ang1
c         algoritmo de Euler-Cromer para os dois osciladores
          vel1 = vel1 - dsin(ang1)*deltat - q*vel1*deltat+
     +           F0*dsin(omegaF*T)*deltat
          ang1 = ang1 + vel1*deltat
          vel2 = vel2 - dsin(ang2)*deltat - q*vel2*deltat+
     +           F0*dsin(omegaF*T)*deltat
          ang2 = ang2 + vel2*deltat
          T = T + deltat
          goto 10
        end if
        write(1, 200)T, '|', ang2-ang1
        close(1)

 100    format(A6, A, A10)
 200    format(F6.2, A, F11.7)

        end
```

Primeiro, observa-se o comportamento do sistema não caótico.

Figura 9: Diferença de ângulo por tempo para $F_0 = 0,5$, escala semilog.



Foi utilizado um intervalo de tempo próximo a 10π, o qual é suficiente para notar o comportamento esperado para esse caso, isto é, $\Delta\theta$ tende a cair conforme o tempo aumenta. Embora o sistema não exiba um gráfico propriamente exponencial, que na escala semi-logarítmica aparentaria uma reta, podemos estimar λ utilizando os 4 pontos que correspondem aos picos observados na Figura 9. Aplicando uma regressão nesses pontos, chegamos em $\lambda = -0,249$, de forma que $\lambda < 0$ evidencia que o sistema é não caótico.

Agora, vejamos como esse parâmetro se comporta no caso caótico.

Figura 10: Diferença de ângulo por tempo para $F_0 = 1,2$, escala semilog.



Note que, dessa vez, foi considerado um intervalo de tempo bem maior, cerca de 200π, a fim de ser possível observar o comportamento divergente de $\Delta\theta$; mesmo que a função apresente alguns mínimos locais, ela tem um comportamento geral visivelmente crescente. Da mesma forma que o caso anterior, embora o gráfico da Figura 10 não apresente um comportamento propriamente exponencial, podemos estimar λ para $t > 100$ s, que deve ser o mais próximo do valor esperado quando $t \to \infty$, pegando a reta média que passa pelos pontos (101.34, 28.7420782) e (628.32, 124.4089841). Portanto, chegamos em $\lambda = 0,00278$, de forma que $\lambda > 0$ evidencia que o sistema é caótico. Observe que o λ encontrado aqui é bem menor, em módulo, do que o valor encontrado para o caso não caótico. Além disso, vale destacar que ambas as estimativas são bastante imprecisas pois foram considerados poucos pontos para o cálculo de cada expoente de *Liapunov*, em relação ao número total de pontos disponìveis.

4 Tarefa D

Na quarta tarefa, foi estudado o comportamento das curvas de $\omega(\theta)$ para os casos vistos no item anterior, isto é, $F_0 = 0,5$ e $F_0 = 1,2$. Com isso, foi possível observar padrões interessantes até mesmo no segundo caso, o qual já foi caracterizado como um movimento caótico; tais padrões não são perdidos nem quando verificamos $\omega(\theta)$ para condições inicias diferentes, porém próximas, algo que será melhor explorado na próxima tarefa, quando o conceito de secção de Poincaré for introduzido.

No programa abaixo, as condições iniciais e as constantes relacionadas ao problema são definidas no início do código, enquanto o algoritmo de Euler-Cromer é implementado logo em seguida, junto com uma parte que garante que o ângulo entre o pêndulo e a vertical esteja sempre entre $-\pi$ e $+\pi$; o código gera ainda um arquivo de saída com os dados utilizados na construção do gráfico para determinadas condições iniciais. Para cada um dos dois valores de F_0, foram testados 3 condições iniciais diferentes: $\theta_0 = 5°$, $\theta_0 = 10°$ e $\theta_0 = 15°$, todas com velocidade angular inicial nula, além dos pontos terem sido coletados até o instante $T = 200\pi$.

tarefa-D-11212550.f:

```
      program tarefaD

        implicit real*8 (a-h,o-z)
        pi = dacos(-1.d0)
c       angulo inicial que o pêndulo faz com a vertical
        ang = pi/12
c       velocidade angular inicial
        vel = 0.d0
c       intervalo utilizado na discretização do tempo
        deltat = 0.03d0
c       tempo total de movimento
        T = 0.d0
c       coeficiente de amortecimento
        q = 0.5d0
c       amplitude de força externa
        F0 = 1.2d0
c       frequência da força externa
        omegaF = 2.d0/3

        open(1, file='saida-D-11212550')
        write(1, 100) 'theta(rad)', '|', 'omega(rad/s)'
 10     if (T.le.200*pi) then
c         algoritmo de Euler-Cromer
          vel = vel - dsin(ang)*deltat - q*vel*deltat+
     +          F0*dsin(omegaF*T)*deltat
          ang = ang + vel*deltat
          T = T + deltat
c         mantém o ângulo sempre entre -pi e +pi
          if (abs(ang).gt.pi) then
            angaux = ang*180/pi
            if (angaux.gt.0) then
              angaux = mod((angaux+180.d0), 360.d0) - 180.d0
            else
              angaux = mod((angaux+180.d0), 360.d0) + 180.d0
            end if
            angaux = angaux*pi/180
            write(1, 200)angaux, '|', vel
          else
```

```
            write(1, 200)ang, '|', vel
          end if
          goto 10
        end if
        close(1)

100     format(A10, A, A12)
200     format(F10.7, A, F11.7)

        end
```

O programa gerou os seguintes resultados para o caso $F_0 = 0,5$:

Figura 11: Velocidade angular por ângulo para $F_0 = 0,5$ e $\theta_0 = 5°$.



Figura 12: Velocidade angular por ângulo para $F_0 = 0,5$ e $\theta_0 = 10°$.



Figura 13: Velocidade angular por ângulo para $F_0 = 0,5$ e $\theta_0 = 15°$.



Note que, nas 3 situações acima, o sistema se comporta da mesma forma, ou seja, inicia próximo da origem do gráfico e evolui até formar uma elipse, e esse padrão de elipse se repete até o fim do intervalo estudado, visto que as linhas da borda dela são mais grossa do que a curva no interior, indicando uma maior concentração de pontos. É possível interpretar esse resultado de acordo com a solução teórica para o problema não caótico, uma vez que ela consiste de um termo transiente e um estacionário; no ínicio, o termo transiente prevalece, então o sistema está evoluindo e não possui uma curva bem definida (varia ligeiramente com a mudança do ângulo inicial), até que o sistema alcança o regime estacionário, representado pelo segundo termo, e os pontos passam a ser distribuídos apenas ao longo da elipse.

Agora, observe como $\omega(\theta)$ se comporta para o caso $F_0 = 1,2$.

Figura 14: Velocidade angular por ângulo para $F_0 = 1,2$ e $\theta_0 = 5°$.



Figura 15: Velocidade angular por ângulo para $F_0 = 1,2$ e $\theta_0 = 10°$.



Figura 16: Velocidade angular por ângulo para $F_0 = 1,2$ e $\theta_0 = 15°$.



Como esperado, os pontos não seguem uma curva bem definida como no caso anterior, pelo contrário, se distribuem de forma mais uniforme no chamado espaço de fase, sendo possível observar verdadeiras manchas, mostrando a grande variedade de regiões que o sistema alcança. Entretanto, destaca-se que o sistema não é tão caótico quanto se esperava, uma vez que, mesmo variando o ângulo inicial, a maior parte da região ocupada pelos pontos não muda. Embora o padrão mais próximo do centro seja mais complicado e não se repete com a variação de θ_0, as regiões mais afastadas praticamente não mudam, sendo possível enxergar uma região bem definida, similar a uma elipse com as extremidades do semi-eixo maior achatadas. Fora dela, não há pontos em nenhum dos 3 casos, mostrando que, mesmo o movimento sendo caótico, há certas regiões que nunca foram visitadas pelo o pêndulo.

5 Tarefa E

Por fim, na última tarefa, que está intimamente relacionada com a anterior, foi estudado o comportamento das curvas de $\omega(\theta)$ para os casos $F_0 = 0,5$ e $F_0 = 1,2$ na secção de Poincaré, isto é, só foram coletados os dados necessários quando, na simulação, o tempo T satisfizer $|T - \frac{n\pi}{\Omega}| < \frac{\Delta t}{2}$, de forma que consideraremos apenas os n pares, o que resultará no gráfico de $\omega(\theta)$ no caso não caótico ser concentrado em um único ponto. No caso caótico, por outro lado, foi possível observar a formação de fractais, conhecidos como atratores estranhos, cuja forma não mesmo variando ligeiramente as condições iniciais.

No programa abaixo, que é bastante similar ao apresentado na tarefa D, ou seja, as condições iniciais e as constantes relacionadas ao problema são definidas no início do código, enquanto o algoritmo de Euler-Cromer é implementado logo em seguida, junto com uma parte que garante que o ângulo entre o pêndulo e a vertical esteja sempre entre $-\pi$ e $+\pi$, com a diferença de que agora, para todo T, calcula-se o n associado e verfica-se ele satisfaz a condição para estar na secção de Poincaré, além de ser par; o código gera ainda um arquivo de saída com os dados utilizados na construção do gráfico para determinadas condições iniciais. Novamente, para o caso $F_0 = 1,2$, foram testados 3 condições iniciais diferentes: $\theta_0 = 5°$, $\theta_0 = 10°$ e $\theta_0 = 15°$, enquanto para o outro foi testada apenas a condição $\theta_0 = 5°$, a fim de verificar se o resultado segue o esperado pela teoria. Em todos os testes, a velocidade angular inicial era nula, além dos pontos terem sido coletados até o instante $T = 600\pi$.

tarefa-E-11212550.f:

```
      program tarefaE

      implicit real*8 (a-h,o-z)
      pi = dacos(-1.d0)
c     angulo inicial que o pêndulo faz com a vertical
      ang = pi/36
c     velocidade angular inicial
      vel = 0.d0
```

```
c       intervalo utilizado na discretização do tempo
        deltat = 0.03d0
c       tempo total de movimento
        T = 0.d0
c       coeficiente de amortecimento
        q = 0.5d0
c       amplitude de força externa
        F0 = 0.5d0
c       frequência da força externa
        omegaF = 2.d0/3

        open(1, file='saida-E-11212550')
        write(1, 100) 'theta(rad)', '|', 'omega(rad/s)'
        write(1, 200)ang, '|', vel
 10     if (T.le.600*pi) then
c         algoritmo de Euler-Cromer
          vel = vel - dsin(ang)*deltat - q*vel*deltat+
     +           F0*dsin(omegaF*T)*deltat
          ang = ang + vel*deltat
          T = T + deltat
          x = omegaF*T/pi
c         encontra o n para traçar a secção de Poincaré
          if ((x-int(x)).le.0.5) then
            n = int(x)
          else
            n = int(x)+1
          end if
c         aplica o critério numérico para os n pares
          if (abs(T -
              n*pi/omegaF).lt.deltat/2.and.mod(n,2).eq.0) then
c           mantém o ângulo sempre entre -pi e +pi
            if (abs(ang).gt.pi) then
              angaux = ang*180/pi
              if (angaux.gt.0) then
                angaux = mod((angaux+180.d0), 360.d0) - 180.d0
              else
                angaux = mod((angaux+180.d0), 360.d0) + 180.d0
              end if
              angaux = angaux*pi/180
              write(1, 200)angaux, '|', vel
            else
              write(1, 200)ang, '|', vel
            end if
          end if
          goto 10
        end if
        close(1)

 100    format(A10, A, A12)
 200    format(F10.7, A, F11.7)
```

```
end
```

O programa gerou os seguintes resultados para o caso $F_0 = 0,5$:

Figura 17: Velocidade angular por ângulo para $F_0 = 0,5$ e $\theta_0 = 5°$ na secção de Poincaré.



Observe que, de fato, a não ser pelo ponto determinado pela condição inicial, todos os outros encontrados se concentram em uma única região, que idealmente deveria ser um único ponto, isto é, se fosse utilizado $\Omega T = n\pi$ ao invés do intervalo considerado, que é mais apropriado para as aproximações numéricas. Vale destacar que o atrator ser um ponto é uma característica de sistemas periódicos, exatamente como o pêndulo se comporta no regime estacionário.

Agora, observe como $\omega(\theta)$ se comporta para o caso $F_0 = 1,2$.

Figura 18: Velocidade angular por ângulo para $F_0 = 1,2$ e $\theta_0 = 5°$ na secção de Poincaré.



Figura 19: Velocidade angular por ângulo para $F_0 = 1,2$ e $\theta_0 = 10°$ na secção de Poincaré.



Figura 20: Velocidade angular por ângulo para $F_0 = 1,2$ e $\theta_0 = 15°$ na secção de Poincaré.



Por fim, o mais interessante: foi visto na tarefa anterior que o sistema, mesmo sendo caótico, possui regiões que nunca são visitadas, e agora, pegando os pontos na secção de Poincaré, é possível observar uma figura mais definida ainda, que pouco muda nos 3 casos, exemplificando a universalidade do caos nesse sistema. O atrator, nesse caso, é então conhecido por atrator estranho, caracterizado por um fractal que é uma estrutura complexa com propriedades bem diferentes, como auto-similaridade e dimensão fracionária. Fractais aparecem sempre como atratores de sistemas caóticos e podem ser usados como um identificador de um sistema desse tipo.